Relation Between Involutions on an Elliptic Curve and the Corresponding Complex Torus This is a question from the book Elliptic Curves by Henry McKean and Victor Moll:

Consider the cubic $X_1: y^2=x^3-x$. It admits the involution $(x,y) \mapsto (\frac{-1}{x}, \frac{y}{x^2})$. It is not the involution $j: (x,y) \mapsto (x,-y)$ so it comes from addition of some half-period. What is that half period?

We know that every automorphism of $\mathbb{C}/\Lambda$ where $\Lambda=\{n_1\omega_1+n_2\omega_2 \mid n_1,n_2\in\mathbb{Z}\}$ for two fixed vectors $\omega_1, \omega_2 \in \mathbb{C}$ is either a translation $f(z)=z+c$ or a glide reflection $f(z)=-z+c$. This can be shown easily by considering $\mathbb{C}$ as the universal cover of $\mathbb{C}/\Lambda$.
I know that I can find the corresponding point of a lattice on an elliptic curve using the $\wp$ function but I can not solve this problem.
Any help is appreciated.
 A: The elliptic curve with equation  $\,y^2=x^3-x\,$ is
Cremona 32a2.
The involution given is
$\,f_3(x,y) := \left(\frac{-1}{x}, \frac{y}{x^2}\right).\,$ Other
involutions are given by
$\,f_2(x,y) := \left(\frac{1-x}{1+x}, \frac{-2y}{(1+x)^2}\right)\,$ and
$\,f_1(x,y) := \left(\frac{x+1}{x-1}, \frac{-2y}{(x-1)^2}\right)\,$ corresponding
to translation by half periods $\,\frac{\omega_3}2,\;
\frac{\omega_2}2,\;\frac{\omega_1}2\,$ respectively.
The real half period $\,\frac{\omega_1}2\,$
is the lemniscate constant
corresponding to the point $\,(1,0).\,$ The imaginary
half period is $\,\frac{\omega_2}2 = -i\,\frac{\omega_1}2\,$
corresponding to the point $\,(-1,0).\,$ The third
half period $\,\frac{\omega_3}2 =  \frac{\omega_1}2+\frac{\omega_2}2\,$
and
corresponds to the point $\,(0,0).\,$
These correspond to the cubic roots $\,e_1=1,\;e_2=-1,\;e_3=0.\,$
My PARI/GP code to verify some of the above is
/* https://www.lmfdb.org/EllipticCurve/Q/32a2/ */
E = ellinit([0,0,0,-1,0]);

[omega1,omega2] = E.omega; omega3 = omega1+omega2;
omegas = [omega1,omega2,omega3];

/* rationalize a complex number */
ba1(x) = bestappr(x, 10^10);
/* rationalize a pair of numbers */
ba2(x) = [ba1(x[1]),ba1(x[2])];

/* should be [e_1,e_2,e_3] = [1,-1,0] */
[e1,e2,e3] = vector(3, n, ba1(ellwp(E,omegas[n]/2))); 
/* should be points [[1,0], [-1,0], [0,0]] */
[p1,p2,p3] = vector(3, n, ba2(ellztopoint(E,omegas[n]/2)));

/* define the three involutions */
f1(p) = my(x,y); [x,y]=p; [(x+1)/(x-1), -2*y/(x-1)^2];
f2(p) = my(x,y); [x,y]=p; [(1-x)/(1+x), -2*y/(1+x)^2];
f3(p) = my(x,y); [x,y]=p; [-1/x, y/x^2];

/* use an example point on curve E */
pa = [2,sqrt(6)];
/* the corresponding z for Weierstrass (p(z),p'(z)) */
za = ellpointtoz(E, pa);
/* should all be [0,0] */
ba2(ellztopoint(E, za + omega1/2) - f1(pa))
ba2(ellztopoint(E, za + omega2/2) - f2(pa))
ba2(ellztopoint(E, za + omega3/2) - f3(pa))

